

Microelectronic Circuits

8th Edition


A. Sedra, K.C. Smith
T. Chan Carusone, V. Gaudet

Spice Problems Solutions
Chapter 8

Prepared by: Nijwm Wary
2019

Problem: 8.5

1. The schematic for this problem is shown below




2. Run the netlist and perform DC simulation. Find out the operating points of the transistors.

NAME	M1	M2
MODEL	NMOS0P18	NMOS0P18
ID	2.50E-05	1.04E-04
VGS	6.79E-01	6.79E-01
VDS	6.79E-01	8.00E-01
VBS	0.00E+00	0.00E+00
VTH	5.00E-01	5.00E-01
VDSAT	1.79E-01	1.79E-01

3. The minimum voltage for the current source M2 to be in saturation region is $V_{CSmin}=V_{OV}$ ($V_{DSAT})= 0.179V$. Also, $I_O=I_{D2}=0.104mA$.

4. Uncomment AC analysis and find out RO by plotting $M(V(VO)/I(V3))$



5. So, the output impedance $R_o= 31.78 k\Omega$.

6. Repeat for other values of VO by changing the dc voltage source V1.

Netlist:

Copy the netlist given below and paste it into a text file and save it with *.cir extension.

```


*****Problem: P8_5 *****
***** Main circuit begins here*****
I1      VDD VG DC 25u
V1      N1 0 0.8Vdc
V2      VDD 0 1.8Vdc
M1      VG VG 0 0 NMOS0P18
+ L=0.54u
+ W=1.74u
+ M=1
M2      VO VG 0 0 NMOS0P18
+ L=0.54u
+ W=6.96u
+ M=1
V3      VO N1 AC 1m
+SIN 0 1m 10k 0 0 0
***** Main circuit ends here*****

***** NMOS model begins here *****
.model NMOS0P18 NMOS(Level=1 VTO=0.5 GAMMA=0.3 PHI=0.84
+ LD=0 WD=0 UO=450 LAMBDA=0.4 TOX=4.08E-9 PB=0.9 CJ=1.6E-3
+ CJSW=2.04E-10 MJ=0.5 MJSW=0.2 CGDO=3.67E-10 JS=8.38E-6 CGBO=3.8E-10
+ CGSO=3.67E-10)
***** NMOS model ends here *****
***** Analysis begins here*****
.OP
*.AC DEC 20 1 100K
.PROBE
.END
***** Analysis ends here*****

```

Problem: 8.45


1. The schematic for this problem is shown below




2. Run the netlist and perform DC simulation. Find out the operating points of the transistors.

NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE
(VI)	.5300	(VO)	.7010	(VDD)	1.3000	(VG23)	.7642


3. It is important to note that V_{OV} is specified at 0.15 V for M1, so V1 should have a dc bias value of 0.55V according to the calculation. This will make M1 close to triode region and so, VO will have limited signal swing. This is because the calculated results in the problem solution do not account for the current flowing due to channel-length modulation (LAMBDA). Because LAMBDA is bigger for the NMOS than the PMOS, it pulls the operating point at the output down towards ground. The effect is quite significant here because L is quite small. So, dc bias of V1 is taken slightly lower 0.53V.
4. Perform transient analysis and plot V(VO) and V(VI). Find the gain of the circuit. The gain is 21.7 V/V.




5. Increase the input swing to 50 mV and perform transient analysis again. Plot $V(VO)$ and the maximum signal swing at the output.



6. The schematic for part (d) of the problem is shown below



7. The output waveform $V(O)$ is shown below



Netlist:

Copy the netlist given below for part (a), (b) and (c). Then paste it into a text file and save it with *.cir extension.

```

*****Problem: P8_45(a,b,c) *****
***** Main circuit begins here*****
M1      VO VI 0 0 NMOS0P13
+ L=0.4u
+ W=7u
+ M=1
M2      VO VG23 VDD VDD PMOS0P13
+ L=0.4u
+ W=27.7u
+ M=1
M3      VG23 VG23 VDD VDD PMOS0P13
+ L=0.4u
+ W=27.7u
+ M=1
I1      VG23 0 DC 100uAdc
V1      VI 0 AC 10m
+SIN 0.53 2m 1k 0 0 0
V2      VDD 0 1.3Vdc
***** Main circuit ends here*****

***** PMOS model begins here *****
.model PMOS0P13      PMOS(Level=1 VTO=-0.4 GAMMA=0.045 PHI=0.8
+                    LD=0 WD=0 UO=100 LAMBDA=0.42 TOX=2.7E-9 PB=0.9)
***** PMOS model ends here *****

***** NMOS model begins here *****
.model NMOS0P13      NMOS(Level=1 VTO=0.4 GAMMA=0.05 PHI=0.8
+                    LD=0 WD=0 UO=400 LAMBDA=0.5 TOX=2.7E-9 PB=0.9)
***** NMOS model ends here *****

***** Analysis begins here*****
.OP
.TRAN 0.01mS 2mS
.PROBE
.END
***** Analysis ends here*****

```

Copy the netlist given below for part (d) and paste it into a text file and save it with *.cir extension.

```

*****Problem: P8_45(d) *****
***** Main circuit begins here*****
M1      VO VI 0 0 NMOS0P13
+ L=0.4u
+ W=7u
+ M=1
V1      VI 0 AC 10m
+SIN 0.53 2m 1k 0 0 0
V2      VDD 0 3.05Vdc
R1      VO VDD 24k TC=0,0
***** Main circuit ends here*****


***** NMOS model begins here *****
.model NMOS0P13      NMOS(Level=1 VTO=0.4 GAMMA=0.05 PHI=0.8
+                    LD=0 WD=0 UO=400 LAMBDA=0.5 TOX=2.7E-9 PB=0.9)
***** NMOS model ends here *****

```


```
***** Analysis begins here*****  
.OP  
.TRAN 0.01mS 2mS  
.PROBE  
.END  
***** Analysis ends here*****
```


Problem: 8.72


1. The schematics for this problem is shown below



2. Run the netlist and perform transient simulation and calculate the gain.



3. The gain is 110 V/V.
4. Find the gain of the common source stage by plotting $V(\text{VD1})$.



5. The gain is 10.7 V/V

Netlist:

Copy the netlist given below and paste it into a text file and save it with *.cir extension.

```

*****Problem: P8_72 *****
***** Main circuit begins here*****
V_DD      VDD 0 1.8Vdc
I1        VDD VOUT1 DC 200uAdc
RL        0 VOUT2 123k
V1        VG2 0 1.421Vdc
C2        VOUT1 VOUT2 10u
Vsig      VSIG 0
+SIN 0.73 10u 1k 0 0 0
M1        VD1 VSIG 0 0 NMOS0P18
+ L=0.36u
+ W=5.4u
+ M=1
M2        VOUT1 VG2 VD1 0 NMOS0P18
+ L=0.36u
+ W=5.4u
+ M=1
***** Main circuit ends here*****


***** NMOS model (0.18um) begins here *****
.model NMOS0P18 NMOS(Level=1 VTO=0.5 GAMMA=0.3 PHI=0.84
+ LD=0 WD=0 UO=450 LAMBDA=0.55 TOX=4.08E-9 PB=0.9 CJ=1.6E-3
+ CJSW=2.04E-10 MJ=0.5 MJSW=0.2 CGDO=3.67E-10 JS=8.38E-6 CGBO=3.8E-10
+ CGSO=3.67E-10)

```

```
***** NMOS model ends here *****  
***** Analysis begins here*****  
.TRAN 0.01mS 2mS  
.PROBE  
.END  
***** Analysis ends here*****
```

Problem: 8.82


1. The schematic for this problem is shown below.




2. Perform the operating point analysis and find the operating point of all transistors.

NAME	M1	M2	M3	M4
MODEL	NMOS0P18	NMOS0P18	NMOS0P18	NMOS0P18
ID	2.00E-05	1.99E-04	1.99E-04	2.00E-05
VGS	6.25E-01	6.25E-01	7.21E-01	7.11E-01
VDS	6.25E-01	6.15E-01	2.85E-01	7.11E-01
VBS	0.00E+00	0.00E+00	-6.15E-01	-6.25E-01
VTH	5.00E-01	5.00E-01	5.87E-01	5.88E-01
VDSAT	1.25E-01	1.25E-01	1.34E-01	1.23E-01

3. Perform DC sweep by uncommenting the command in the analysis section and plot the I(V2). Note the output voltage at which the output current becomes constant.



4. Perform AC analysis by uncommenting the command in the analysis section and plot the $M(V(VOUT)/I(U2))$ to calculate the output resistance.



Netlist:

Copy the netlist given below and paste it into a text file and save it with *.cir extension.

```

*****Problem: P8_82 *****
***** Main circuit begins here*****
M1      VG12 VG12 0 0 NMOS0P18
+ L=0.54u
+ W=2.7u
+ M=1
M2      N14530 VG12 0 0 NMOS0P18
+ L=0.54u
+ W=27u
+ M=1
M3      VOUT VG34 N14530 0 NMOS0P18
+ L=0.54u
+ W=27u
+ M=1
M4      VG34 VG34 VG12 0 NMOS0P18
+ L=0.54u
+ W=2.7u
+ M=1
I1      VDD VG34 DC 20uAdc
V1      VDD 0 1.8Vdc
V2      VOUT 0 AC 1k
+SIN 0.9 1m 1k 0 0 0
***** Main circuit ends here*****

***** NMOS model (0.18um) begins here *****
.model NMOS0P18      NMOS(Level=1 VTO=0.5 GAMMA=0.3 PHI=0.84
+          LD=0 WD=0 UO=450 LAMBDA=0.55 TOX=4.08E-9 PB=0.9 CJ=1.6E-3
+          CJSW=2.04E-10 MJ=0.5 MJSW=0.2 CGDO=3.67E-10 JS=8.38E-6 CGBO=3.8E-10
+          CGSO=3.67E-10)
***** NMOS model ends here *****

***** Analysis begins here*****
.OP
*.DC [LIN] V2 0.3 1.0 0.05
*.AC DEC 20 1 100K
.PROBE
.END
***** Analysis ends here*****

```